/*
Copyright (c) 2009, Young Digital Planet SA
All rights reserved.

* Redistribution and use in source and binary forms, with or without 
  modification, are permitted provided that the following conditions are met:
* Redistributions of source code must retain the above copyright notice, this 
  list of conditions and the following disclaimer.
* Redistributions in binary form must reproduce the above copyright notice, 
  this list of conditions and the following disclaimer in the documentation 
  and/or other materials provided with the distribution.
* Neither the name of the Young Digital Planet SA nor the names of its 
  contributors may be used to endorse or promote products derived from this 
  software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE 
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE 
LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR 
CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF 
SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS 
INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN 
CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) 
ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE 
POSSIBILITY OF SUCH DAMAGE.
*/
﻿package pl.ydp.utils
{
	import flash.display.DisplayObject;
	import flash.display.DisplayObjectContainer;
	import flash.utils.ByteArray;
	
	/**
	 * Narzędzia do operowania na obiektach-map'ach
	 */
	public class ObjectUtil
	{
		/**
		 * Tworzy obiekt zawierający różnicę w properties. 
		 * @param base podstawia do określenia zmian
		 * @param nuObject obiekt zaiwerający nowe ustawienia
		 * @return obiekt, który zawiera tylko properties zmodyfikowane w
		 * nuObject. Jeżeli jakieś property zostało usunięte to obiekt
		 * będzie zawierał null. <b>Gdy obiekty mają równy zestaw properties
		 * zwracany jest null.</b>
		 */
		public static function diff( base:Object, nuObject:Object ):Object {
			var diff:Object = { };
			var count:uint = 0;
			
			for( var n:String in nuObject ) {
				var old:Object = base[n];
				var nu:Object = nuObject[n];
				if ( old!=nu ) {
					diff[n] = nu;
					++count;
				}
			}
			for( n in base ) {
				if ( nuObject[n]===undefined ) {
					diff[n] = null;
					++count;
				}
			}
			
			return count==0 ? null : diff;
		}
		
		/**
		 * Porównuje objekty traktując je jako mapę. Sprawdza czy 
		 * obydwa obiekty posiadają identyczne properties i czy
		 * wartości pod tymi properties nie sa różne (!=).
		 * @param base bazowy obiekt porównania
		 * @param nuObject derugi obiekt porównania
		 * @return true gdy obiekty są takie same
		 */
		public static function objEqual( base:Object, nuObject:Object ):Boolean {
			var eq:Boolean = true;
			for( var n:String in nuObject ) {
				var old:Object = base[n];
				var nu:Object = nuObject[n];
				if ( old!=nu ) {
					eq = false;
					break;
				}
			}
			if ( eq ) {
				for( n in base ) {
					old = base[n];
					nu = nuObject[n];
					if ( old!=nu ) {
						eq = false;
						break;
					}
				}
			}
			
			return eq;
		}

		/**
		 * Wyświetla rekrusyjnie listę dzieci podanego kontenera
		 * @param container container do wyświetlenia
		 * @param indentString indent (zwiększany z każdym poziomem)
		 */
		public static function traceDisplayList(container:DisplayObjectContainer,indentString:String = ""):void
		{
		    var child:DisplayObject;
		    for (var i:uint=0; i < container.numChildren; i++)
		    {
		        child = container.getChildAt(i);
		        trace (indentString, child, child.name); 
		        if (container.getChildAt(i) is DisplayObjectContainer)
		        {
		            traceDisplayList(DisplayObjectContainer(child), indentString + "    ")
		        }
		    }
		}
		
		/**
		 * Kopiuje obiekt older, dodaje do niego własności obiektu newer, 
		 * nadpisując stare wartości i zwraca obiekt wynikowy
		 * @param older baza kopiowania
		 * @param newer zawiera property, które mają zostać dołączone/zupdate'owane
		 * @return obiekt zawierający property z older i z newer
		 */
		public static function merge(older:Object,newer:Object):Object {
			var merged:Object = (older==null)?
				new Object() : ObjectUtil.copy( older );
			for (var n:String in newer) {
				merged[n] = newer[n];
			}
			return merged;
		}
		
		/**
		 * Kopiuje obiekt i zwraca refencje do nowo utworzonej kopii, 
		 * @param value obiekt do skopiowania
		 * @return referencja do nowego obiektu 
		 */
		public static function copy(value:Object):Object{
	        var buffer:ByteArray = new ByteArray();
	        buffer.writeObject(value);
	        buffer.position = 0;
	        var result:Object = buffer.readObject();
	        return result;
    	}
		
		/**
		 * Sprawdza czy podany obiekt ma jakiekolwiek pole.
		 * @param value obiekt do zbadania
		 * @return zwraca true jeśli obiekt jest pusty, w przeciwnym przypadku false.
		 */
		public static function isEmpty(value:Object):Boolean{
			var propsCnt:uint = 0;
			
			for(var prop:String in value){
				propsCnt = 1;
				break;
			}
			
			return propsCnt == 0;
		}

	}
}